

Lesson 11

Trigonometric Substitutions

Initializations

```
> restart;  
with(student);  
>
```

11.1 Trigonometric Substitutions

Integrals involving $\sqrt{x^2 \pm a^2}$ can be evaluated using a trigonometric substitution which turns the expression under the radical into a perfect square. Let $a > 0$.

- For $\sqrt{a^2 - x^2}$: Choose $x = a \sin u$ $-\frac{\pi}{2} \leq u \leq \frac{\pi}{2}$
- For $\sqrt{x^2 + a^2}$: Choose $x = a \tan u$ $-\frac{\pi}{2} < u < \frac{\pi}{2}$
- For $\sqrt{x^2 - a^2}$: Choose $x = a \sec u$ $0 \leq u < \frac{\pi}{2}$ or $\pi \leq u < \frac{3\pi}{2}$

Examples

Example 11.1.1

Evaluate $\int \frac{1}{x^2 \sqrt{x^2 + 9}} dx$.

Solution

Since the integrand involves $\sqrt{x^2 + 9}$ the substitution $x = 3 \tan u$ is in order. The **isolate** command allows us to easily obtain the inverse transformation.

```
> e1:=Int(1/(x^2*sqrt(x^2+9)), x);
```

$$e1 := \int \frac{1}{x^2 \sqrt{x^2 + 9}} dx \quad (2.1.1.1)$$

```
> rr:=x=3*tan(u);  
inv_rr:=isolate(rr, u);
```

$$rr := x = 3 \tan(u)$$

$$inv_rr := u = \arctan\left(\frac{1}{3} x\right) \quad (2.1.1.2)$$

Perform the substitution.

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int \frac{1}{3} \frac{1 + \tan(u)^2}{\tan(u)^2 \sqrt{9 \tan(u)^2 + 9}} du \quad (2.1.1.3)$$

Use the **simplify** command with the **symbolic** option to get the integrand into the desired form.

```
> e3:=simplify(e2, symbolic);
```

$$e3 := \frac{1}{9} \int \frac{\cos(u)}{\sin(u)^2} du \quad (2.1.1.4)$$

This integral can be solved by means of a regular u -substitution. We simply ask for its value.

```
> e4:=value(e3)+C;
```

$$e4 := -\frac{1}{9 \sin(u)} + C \quad (2.1.1.5)$$

Finally, translate this result back in terms of the variable x .

```
> e5:=subs(inv_rr, e4);
```

$$e5 := -\frac{1}{9 \sin\left(\arctan\left(\frac{1}{3} x\right)\right)} + C \quad (2.1.1.6)$$

```
> e6:=expand(e5);
```

$$e6 := -\frac{1}{9} \frac{\sqrt{x^2+9}}{x} + C \quad (2.1.1.7)$$

```
>
```